AD-A081 328   EPL ANALYSIS OLNEY MD   F/G 15/4
MATHEMATICAL MODEL OF A MIXED SURVEILLANCE SYSTEM.(U)
DEC 77 P S SHOENFELD

UNCLASSIFIED   NL

1 OF 1
AD
A081328

END
DATE
FILMED
4-80
DDC

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE

ADA081328

DDC FILE COPY

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER
2. GOVT ACCESSION NO.
3. RECIPIENT'S CATALOG NUMBER

4. TITLE (and Subtitle)
Mathematical Model of a Mixed Surveillance System. (U)

5. TYPE OF REPORT & PERIOD COVERED
Memorandum rept.

6. PERFORMING ORG. REPORT NUMBER

7. AUTHOR(s)
Peter S. Shoenfeld

8. CONTRACT OR GRANT NUMBER(s)

9. PERFORMING ORGANIZATION NAME AND ADDRESS
EPL Analysis
Olney, MD 20702

10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS

11. CONTROLLING OFFICE NAME AND ADDRESS
CNO (OP-96)
Washington, D.C. 20350

12. REPORT DATE
29 December 1977

13. NUMBER OF PAGES
31

14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office)
12 31

15. SECURITY CLASS. (of this report)
UNCLASSIFIED

15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Unlimited and approved for Public release.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

DTIC ELECTE FEB 28 1980 S D A

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Advanced Naval Vehicle Concepts Evaluation
ANVCE
Military Worth
Surveillance System

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

This memorandum describes a model of reactive surveillance in which two classes of contacts occur. A single search vehicle attempts to maintain some localization of a single target vehicle with contacts occurring intermittently as a Poisson process but only when the target is within a certain range (exposure disk) of the searcher. An external surveillance system is also present; it produces contacts as a Poisson process irrespective of the target position. Searcher tactics are assumed to approximate optimal

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

409 403

search based on an assumed cir ... mal target distribution determined solely by the class ... rcher or external system) of the most recent contact and time since that contact.

gp-024 L

EPL Analysis

CONSULTANTS IN

APPLIED MATHEMATICS, OPERATIONS RESEARCH

OWARD P. LOANE
TER S. SHOENFELD
(301) 924-3574

SUITE 212
3460 OLNEY-LAYTONSVILLE ROAD
OLNEY, MARYLAND 20832

December 29, 1977

MEMORANDUM

To: ANVCE (OP-96V)
Department of the Navy
Arlington, Virginia
Attn: Dr. H. L. Weiner

From: Peter S. Shoenfeld

Subject: Mathematical Model of a Mixed Surveillance System

This memorandum describes a model of reactive surveillance in which two classes of contacts occur. A single search vehicle attempts to maintain some localization of a single target vehicle with contacts occurring intermittently as a Poisson process but only when the target is within a certain range (exposure disk) of the searcher. An external surveillance system is also present; it produces contacts as a Poisson process irrespective of the target position. Searcher tactics are assumed to approximate optimal search based on an assumed circular normal target distribution determined solely by the class (searcher or external system) of the most recent contact and time since that contact.

The remainder of this memorandum is divided into sections. The mathematical model and possible generalizations are described in the first section. Numerical techniques employed for the evaluation of limits and integrals are discussed in the second. The computational algorithm is summarized in the third section. A computer program listing is included as an appendix.

Mathematical Model

Three types of contacts are considered; those by the searcher, those by the external system when the target lies within the searcher's exposure disk, and those by the external system when the target lies outside this disk. Probabilistically, the sequence of contact types and corresponding contact times is described as a Markov renewal process in which the distribution of next contact type and recontact time depends only on the type of most recent contact; the process regenerates with each contact. The sequence of contact types is a Markov chain. The intervals between contacts are described separately as inhomogeneous, continuous time parameter, Markov processes with different time dependent transition functions and initial probability vectors according to the type of most recent contact. These processes have five states corresponding to the three types of contacts (absorbing states) and target locations within and outside of the searcher's exposure disk before recontact. The assumed circular normal target distribution is described by a linear datum growth law with initial datum area determined by contact type and datum growth rate determined by target motion statistics.

These concepts are intended to model reactive surveillance by ANVs; the external contacts might correspond to those generated by SOSUS or by all other surveillance systems combined. However, the concepts generalize to situations involving differing numbers of contact types, transition functions, and datum growth laws.

The principal measure of effectiveness derived in this memorandum is distribution of actual target range from the expected (by the searcher) target location at a random time instant. This has obvious application to anti-SLBM defense. Many other such measures could be derived from the information generated by the model.

The five states in the interval following a contact are numbered and described as follows:

State 1 - Recontact by the external surveillance system has occurred while the target is exposed to contact by the searcher.

State 2 - Recontact by the external system has occurred while the target is unexposed to the searcher.

State 3 - Recontact by the searcher has occurred.

State 4 - Recontact has not yet occurred and the target is exposed.

State 5 - Recontact has not yet occurred and the target is unexposed.

The three types of contacts have the same numbering as the first three states.

Let

$$P_{ij}(t) = \text{Probability}\left(\begin{array}{l}\text{process in}\\ \text{state } i\\ \text{at time } t\\ \text{after last}\\ \text{contact}\end{array}\middle|\begin{array}{l}\text{last}\\ \text{contact of}\\ \text{type } j\end{array}\right)\left(\begin{array}{l}i = 1,2,3,4,5\\ j = 1,2,3\end{array}\right)$$

and

$$P^j(t) = (P_{1j}(t),\cdots,P_{5j}(t)), \quad (j = 1,2,3)$$

where

$$t = \text{time since last contact.}$$

Three separate Markov processes are obtained for $j = 1,2,3$. By definition,

$$P^1(0) = (0,0,0,1,0)$$

and

$$P^2(0) = (0,0,0,0,1).$$

Since the target must be exposed for contact by the searcher to occur,

$$P^3(0) = (0,0,0,1,0).$$

The model uses three absorbing states corresponding to contact types although the use of only two (external and by searcher) would appear more natural. This is done so that the initial probability vectors, $P^j(0)$, can be specified a priori. In adapting the model to a more general situation in which there are N (in this case two) classes of contacts and M (in this case also two) intervening states, as many as MN (in this case three sufficed) absorbing states might be required.

Let

$$\xi(A) = \begin{bmatrix} \text{unconditional probability} \\ \text{that target is exposed to} \\ \text{searcher during search of} \\ \text{datum area A} \end{bmatrix}$$

and

$$\eta(A) = \begin{bmatrix} \text{rate at which unexposed targets} \\ \text{enter exposure disk during} \\ \text{search of datum area A.} \end{bmatrix}$$

$$= \frac{\partial}{\partial \tau} \left\{ \text{Probability} \begin{bmatrix} \text{target exposed} \\ \text{at time } t+\tau \end{bmatrix} \middle| \begin{matrix} \text{target} \\ \text{unexposed} \\ \text{at time } t \end{matrix} \right\} \Bigg|_{\tau=0}.$$

"Datum area" is defined by

$$A = 6\pi\sigma^2 \tag{1}$$

where σ^2 is the parameter characterizing the assumed circular normal target distribution. This is the area of a bounded region which, if it contained the target with a uniform location distribution, would require the same expected effort to detect as is required by a circular normal distribution with parameter σ^2 (see reference [a]). While in actuality the quantities ξ and ψ depend on geographic details and are not functions of A alone, the following formulas are reasonable approximations:

$$\xi(A) = 1 - \exp(-\pi R_E{}^2/A) \tag{2}$$

and

$$\psi(A) = 2R_E V_s/A, \tag{3}$$

where

R_E = radius of exposure disk, and

V_s = searcher patrol speed.

Formula (2) subsumes searcher allocation of effort and non-regularity in the search region. Formula (3) is familiar and standard for $\pi R^2 \ll A$, it is shown below that it is sensible for small A as well.

Consider an arbitrary R_E and A. If the target is <u>not</u> exposed to detection, it is confined to a region of expected area

$$A \exp\left(\frac{-\pi R_E^2}{A}\right).$$

Let $\tilde{S}$ be the speed at which the boundary of the disc of exposure sweeps through this unexposed area and let $\tilde{W}$ be effective sweep width. For $\pi R_E^2 \ll A$, the entire boundary intersects A, $\tilde{S} = V_s$ and $\tilde{W} \approx 2R_E$. For $\pi R_E^2 \gg A$, only a small portion of the boundary intersects A, and speed perpendicular to that segment is $V_s(2/\pi)$. But with probability .5, that segment is sweeping into exposed area, i.e., no new area is exposed. Thus $\tilde{S} = S/\pi$ and $\tilde{W} \approx$ the length of boundary intersecting A.

Now if the target is located in the unexposed area, the rate of entry into the disc of exposure is

$$\frac{\tilde{S}\tilde{W}}{A \exp(-\pi R_E^2/A)}.$$

Equating this to (3), one has

$$\tilde{W} = \frac{2R_E V_s}{\tilde{S}} \exp\left(\frac{-\pi R_E^2}{A}\right) \tag{4}$$

which, for $\pi R_E^2 \ll A$, gives the expected result, $\tilde{W} \approx 2R_E$. As A decreases in size, the right-hand member of (4) decreases, and can be bounded by

$$\tilde{W} \le \left(\frac{V_s}{\tilde{S}}\right)\left(\frac{A}{\pi}\right)^{\frac{1}{2}} (\sqrt{2}\exp(-.5)).$$

Taking $\tilde{S} = S/\pi$, a presumed minimum value,

$$\tilde{W} \le \left(\frac{A}{\pi}\right)^{\frac{1}{2}} (2.694),$$

a very reasonable result. Thus formula (3) appears sensible for small A, despite the inverse power dependence.

The assumed linear datum growth law after contacts of type j is

$$A_j(t) = A_j(0) + \rho t \tag{5}$$

where

$A_j(t)$ = the datum area at a time t after a contact of type j, and

ρ = datum growth rate,

for j = 1,2,3.

Since type 1 and type 2 contacts both come from the same external surveillance system, the resulting initial datum is the same for each ($A_1(0) = A_2(0)$). The growth rate ρ is determined by target motion statistics. If the target is moving in an unbiased random walk with

V_t = target speed, and

C = mean time between target course changes,

it may be shown that the target location distribution is approximately circularly normal with density in polar coordinates (r,θ),

$$f(r,\theta) = \frac{1}{2\pi\sigma^2(t)} \exp\left(\frac{-r^2}{2\sigma^2(t)}\right) \tag{6}$$

where

$$\sigma^2(t) = V_t^{\,2}\, C^2\, [t/C - (1 - \exp(-t/C))].$$

Thus for large t,

$$\frac{\partial \sigma^2(t)}{\partial t} \approx V_t^{\,2}\, C$$

and, since $A = 6\pi\sigma^2$,

$$\rho = 6\pi\, V_t^{\,2}\, C \,. \tag{7}$$

Formula (7) is used to compute ρ in applications.

Let

α = external recontact rate

$$= \frac{\partial}{\partial \tau}\left\{ \text{Probability}\left[\begin{array}{l}\text{external}\\ \text{recontact by}\\ \text{time } t+\tau\end{array}\middle|\begin{array}{l}\text{no recontact}\\ \text{by time } t\end{array}\right]\right\}\Bigg|_{\tau=0},$$

λ = searcher recontact rate for an exposed target

$$\frac{\partial}{\partial \tau}\left\{ \text{Probability}\left[\begin{array}{l}\text{searcher}\\ \text{recontact by}\\ \text{time } t+\tau\end{array}\middle|\begin{array}{l}\text{target exposed}\\ \text{at time } t \text{ and}\\ \text{no recontact}\\ \text{by time } t\end{array}\right]\right\}\Bigg|_{\tau=0},$$

$\beta_j(t)$ = rate at which unexposed targets enter the exposure disk after a type j contact

$$= \frac{\partial}{\partial\tau}\left\{\text{Probability}\left[\begin{array}{l}\text{target}\\ \text{exposed at}\\ \text{time } t+\tau\end{array}\middle|\begin{array}{l}\text{target unexposed}\\ \text{at time t and}\\ \text{last contact of}\\ \text{type j}\end{array}\right]\right\}\Bigg|_{\tau=0},$$

and

$\gamma_j(t)$ = rate at which exposed targets leave exposure disk

$$= \frac{\partial}{\partial\tau}\left\{\text{Probability}\left[\begin{array}{l}\text{target}\\ \text{unexposed at}\\ \text{time } t+\tau\end{array}\middle|\begin{array}{l}\text{target exposed}\\ \text{at time t and}\\ \text{last contact of}\\ \text{type j}\end{array}\right]\right\}\Bigg|_{\tau=0},$$

where

t = time since last contact,

for j = 1,2,3.

It is assumed that the rates α and λ are actually constant and that the rates β_j and γ_j are functions of t alone. The model requires these assumptions, which seem quite reasonable. It follows that

$$\frac{dP^j(t)}{dt} = P^j(t)\begin{bmatrix} 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ \alpha & 0 & \lambda & -[\alpha+\lambda+\gamma_j(t)] & \gamma_j(t) \\ 0 & \alpha & 0 & \beta_j(t) & -[\alpha+\beta_j(t)] \end{bmatrix}. \qquad (8)$$

Now

$$\beta_j(t) = \eta(A_j(t))$$

$$= \frac{2R_E V_s}{A_j(t)} \tag{9}$$

by formula (3). For the case $\alpha = \lambda = 0$, $P_{4j}(t) = \xi(A_j(t))$ and the fourth equation in (8) becomes

$$\xi'(A_j(t))A_j'(t) = -\gamma_j(t)\xi(A_j(t)) + \beta_j(t)(1 - \xi(A_j(t))).$$

Given a form for ξ this can be solved for γ_j in terms of β_j and A_j. Using formula (2) for ξ,

$$\gamma_j(t) = \left[\frac{\exp(-\pi R_E^2/A_j(t))}{1 - \exp(-\pi R_E^2/A_j(t))}\right]\left[\beta_j(t) + \frac{\pi R_E^2 \frac{\partial A_j(t)}{\partial t}}{[A_j(t)]^2}\right]$$

$$= \left[\frac{\exp(-\pi R_E^2/A_j(t))}{1 - \exp(-\pi R_E^2/A_j(t))}\right]\left[\beta_j(t) + \frac{\pi R_E^2 \rho}{[A_j(t)]^2}\right]. \tag{10}$$

System (8) could be easily modified to discard external contacts occurring at times when their information is inferior to that already available from the last contact. This would entail replacing the term α in (8) by

$$\alpha_j(t) = \begin{Bmatrix} 0 & \text{if} \quad A_j(t) < A_1(0) \\ \alpha & \text{otherwise} \end{Bmatrix} .$$

This modification would have little effect in most cases.

Let

$$B_{ij} = \lim_{t \to \infty} P_{ij}(t) \quad \text{for } i,j = 1,2,3.$$

It may be shown that since $\alpha > 0$, the matrix $[B_{ij}]$ is well defined, has all positive entries, and all column sums one. $[B_{ij}]$ is the transition matrix of the Markov chain characterizing the sequence of contact types. It has a unique eigenvector summing to one with eigenvalue one; this is the vector of steady state probabilities. Let

$$(\theta_1,\theta_2,\theta_3) = \left[\begin{array}{l}\text{steady state probability vector for} \\ \text{Markov chain characterizing sequence} \\ \text{of contact types}\end{array}\right]. \qquad (11)$$

If

$$\Phi_1 = \frac{B_{13}(B_{22}-1) - B_{23}B_{12}}{(B_{11}-1)(B_{22}-1) - B_{12}B_{21}},$$

$$\Phi_2 = \frac{B_{23}(B_{11}-1) - B_{13}B_{21}}{(B_{11}-1)(B_{22}-1) - B_{12}B_{21}}, \quad \text{and}$$

$$\Phi_3 = -1$$

then (12)

$$\theta_j = \frac{\Phi_j}{\Phi_1 + \Phi_2 + \Phi_3} \quad \text{for } j = 1,2,3.$$

Define

$$Q_j(t) = \text{Probability}\left[\text{recontact within time } t \,\middle|\, \text{last contact of type } j\right]$$

and

$$\Omega_j(t) = \int_0^t [1 - Q_j(s)]\, ds \,. \tag{13}$$

Then

$$Q_j(t) = \sum_{i=1}^{3} P_{ij}(t) \,. \tag{14}$$

Integrating by parts,

$$\begin{bmatrix} \text{Conditional expectation} \\ \text{of duration of portion} \\ \text{of interval between} \\ \text{contacts where time} \\ \text{since last contact} < t \\ \text{given that last contact} \\ \text{of type } j \end{bmatrix} = \int_0^\infty \begin{Bmatrix} s & \text{if} & s \le t \\ t & \text{if} & s \ge t \end{Bmatrix} Q'(s)\, ds$$

$$= \Omega_j(t). \tag{15}$$

In particular, defining

$$L_j = \lim_{t \to \infty} \Omega_j(t), \tag{16}$$

$$L_j = \begin{bmatrix} \text{Expected value of} \\ \text{time until recontact} \\ \text{after contact of} \\ \text{type } j \end{bmatrix}. \tag{17}$$

This limit certainly exists since $\alpha > 0$ and (8) imply that $1 - Q_j(t)$ is bounded above by $\exp(-\alpha t)$.

The next derivation assumes that the Markov chain of contact types has reached its steady state; in reality this condition and its consequences are approached asymptotically for long search histories. Let

$$l_j = \begin{bmatrix} \text{Expected value of} \\ \text{time between contact} \\ \text{of type j and last} \\ \text{preceding contact} \end{bmatrix}. \tag{18}$$

Then, by Bayes Theorem,

$$l_j = \sum_{k=1}^{3} L_k \text{ Probability} \left[\begin{matrix} \text{preceding} \\ \text{contact of} \\ \text{type k} \end{matrix} \middle| \begin{matrix} \text{contact of} \\ \text{type j} \end{matrix} \right]$$

$$= \sum_{k=1}^{3} L_k \left\{ \frac{B_{jk}\theta_k}{\sum_{i=1}^{3} B_{ji}\theta_i} \right\}$$

$$= \frac{1}{\theta_j} \sum_{k=1}^{3} L_k B_{jk} \theta_k. \tag{19}$$

Also, defining

$$\mathcal{I} = \begin{bmatrix} \text{unconditioned expected value} \\ \text{of time between contacts} \end{bmatrix}, \tag{20}$$

$$\mathcal{I} = \sum_{k=1}^{3} \theta_k L_k . \tag{21}$$

Consider a long interval of search history of duration T_M

for which the initial state is assumed known*. Let

$$T = \begin{bmatrix} \text{a random time instant, uniformly} \\ \text{distributed on } [0,T_M] \end{bmatrix},$$

$$j_1 = \begin{bmatrix} \text{the type of the last contact preceding} \\ \text{T (0 if there is no such contact)} \end{bmatrix},$$

$$j_2 = \begin{bmatrix} \text{the type of the first contact following} \\ \text{T (0 if there is no such contact)} \end{bmatrix}, \text{ and}$$

$$T_e = \begin{bmatrix} \text{time elapsed at T since last contact} \\ \text{(0 if there is no such contact)} \end{bmatrix}.$$

The random variables j_1, j_2, and T_e all have limiting distributions as $T_M \to \infty$ which are independent of the initial state. These distributions are:

$$\text{Probability } [j_1 = j] = \frac{\theta_j L_j}{L} \quad \left(\begin{matrix}\text{for } j = 1,2,3; \\ \text{by (11), (17), and (20)}\end{matrix}\right), \tag{22}$$

$$\text{Probability } [j_2 = j] = \frac{\theta_j l_j}{L} \quad \left(\begin{matrix}\text{for } j = 1,2,3; \\ \text{by (11), (18), and (20)}\end{matrix}\right), \tag{23}$$

and

$$\text{Probability } [T_e < t] = \frac{\sum_{k=1}^{3} \theta_k \Omega_k(t)}{L} \quad \left(\begin{matrix}\text{for } 0 \le t < \infty; \\ \text{by (11), (15), and (20)}\end{matrix}\right). \tag{24}$$

Now define

$$F(Y) = \text{Probability} \begin{bmatrix} \text{datum area } A < Y \\ \text{at random time } T \end{bmatrix}, \text{ and}$$

$$A_j^{-1}(Y) = \begin{cases} 0 & \text{if } Y \le A_j(0) \\ \text{unique } t & \text{such that } A_j(t) = Y \text{ otherwise} \end{cases}.$$

----

* Here the model is regarded as a five state process with continuous time parameter in which the three states corresponding to contacts are attained only for discrete instants.

Then by (24),

$$F(Y) = \frac{\sum_{k=1}^{3} \theta_k \Omega_k(A_k^{-1}(Y))}{\Sigma} . \tag{25}$$

With an assumed circular normal datum with parameter σ^2, formula (6) implies

$$\text{Probability} \begin{bmatrix} \text{target within} \\ \text{distance R of} \\ \text{datum center} \end{bmatrix} = 1 - \exp(-R^2/2\sigma^2).$$

Defining

$$G(R) = \begin{bmatrix} \text{Unconditional steady state probability} \\ \text{that target is within distance R of} \\ \text{datum center} \end{bmatrix},$$

$$G(R) = \int_0^{\infty} [1 - \exp(-R^2/2x)] \frac{dF(6\pi x)}{dx} dx \text{ , using (1).}$$

Integrating by parts and changing variables,

$$G(R) = 3\pi R^2 \int_0^{\infty} \frac{\exp(-3\pi R^2/U) F(U)}{U^2} dU .$$

Using formulas (25) and (5) for the functions F and A_j and changing variables again leads to

$$G(R) = \frac{3\pi\rho R^2}{\Sigma} \sum_{k=1}^{3} \left\{ \theta_k \int_0^{\infty} \frac{\exp[-3\pi R^2/(A_k(0) + \rho t)]}{(A_k(0) + \rho t)^2} \Omega_k(t)\, dt \right\} . \tag{26}$$

Evaluation of Limits and Integrals

System (8) is evaluated numerically by the Runge-Kutta method for $j = 1,2,3$. The vectors $P^j(t) = (P_{1j}(t),\cdots,P_{5j}(t))$ and $dP^j/dt = (P_{1j}'(t),\cdots,P_{5j}'(t))$ are available explicitly at each increment. The evaluation is carried out over the interval $[0, T_{max}]$, where T_{max} is chosen so that the recontact probability, $Q_j(T_{max})$, will be quite large (say $\geq .95$) for each j. The contact type transition probabilities are

$$B_{ij} = \lim_{t\to\infty} P_{ij}(t).$$

Since

$$\sum_{i=1}^{3} B_{ij} = 1$$

necessarily, a convenient and reasonable approximation is

$$B_{ij} \sim P_{ij}(T_{max}) + \frac{P_{ij}'(T_{max})}{\sum_{i=1}^{3} P_{ij}'(T_{max})} \left(1 - \sum_{i=1}^{3} P_{ij}(T_{max})\right) . \tag{27}$$

The functions $\Omega_j(t)$ were defined as

$$\Omega_j(t) = \int_0^t [1 - Q_j(s)]\, ds$$

where $Q_j(s) = \sum_{i=1}^{3} P_{ij}(s)$. Since values of the P_{ij} and their derivatives are explicitly available at fine increments from the Runge-Kutta evaluation of system (8), the Ω_j can be obtained at the same time to reasonable accuracy by trapezoidal integration for times in the interval $[0, T_{max}]$. The limits

$$L_j = \lim_{t \to \infty} \Omega_j(t)$$

are also required. Approximations to the L_j are obtained by assuming that the probability of no recontact becomes a negative exponential after a long time. This assumption is reasonable, particularly if datum growth really ceases after a long time. More precisely it is assumed that for $t \geq T_{max}$,

$$\frac{1 - Q_j(t)}{1 - Q_j(T_{max})} = \exp[-\psi_j(t - T_{max})]$$

for some constant ψ_j. This leads to the approximation

$$L_j \approx \Omega_j(T_{max}) + \frac{[1 - Q_j(T_{max})]^2}{Q_j'(T_{max})}, \tag{28}$$

which is calculable since Ω_j, Q_j, and Q_j' are all available for $t = T_{max}$.

The integrals in

$$G(R) = \frac{3\pi\rho R^2}{\Gamma} \sum_{k=1}^{3} \left\{ \theta_k \int_0^{\infty} \frac{\exp[-3\pi R^2/(A_k(0) + \rho(t)]}{(A_k(0) + \rho t)^2} \Omega_k(t)\, dt \right\}$$

must be evaluated for a number of values of R. The portion of each integral in the interval $[0, T_{max}]$ is evaluated from a table of values of $(t, \Omega_k(t))$ which was saved from the Runge-Kutta integration. The increment size changes several times in the table. Simpson's rule is used separately on each interval with a constant increment size; trapezoidal integration is used wherever an odd subinterval is left over. These integrals have substantial "tails" in the interval $[T_{max}, \infty]$. These are estimated by assuming that for $t > T_{max}$,

$$\Omega_k(t) \sim L_k, \text{ and}$$

$$A_k(0) + \rho t \sim \rho t.$$

This leads to the approximation

$$G(R) \sim \frac{3\pi\rho R^2}{L} \sum_{k=1}^{3} \left\{ \theta_k \int_0^{T_{max}} \frac{\exp[-3\pi R^2/(A_k(0)+\rho t)]}{(A_k(0)+\rho t)^2} \Omega_k(t)\, dt \right\}$$

$$+ \left\{ 1 - \exp(-3\pi R^2/\rho T_{max}) \right\} . \qquad (29)$$

Summary of Computational Algorithm

The algorithm is embodied in a computer program which is included as Appendix A. Inputs are:

- V_s = searcher patrol speed (nm./hr.),
- R_E = radius of exposure disk (nm.),
- α = external recontact rate (hr.$^{-1}$) when target is exposed,
- λ = searcher recontact rate (hr.$^{-1}$),
- ρ = datum growth rate (nm.^{2}/hr.),
- $A_1(0)$ = initial datum size after external contacts (nm.^{2}), and
- $A_3(0)$ = initial datum size (nm.^{2}) after searcher recontacts.

The systems of linear first-order differential equations (8) are integrated over the interval $[0, T_{max}]$ by a library Runge-Kutta routine for $j = 1,2,3$. The rates $\beta_j(t)$ and $\gamma_j(t)$ are calculated by formulas (9) and (10) using the datum growth law (5). Values of $Q_j(t)$ and $\Omega_j(t)$ are calculated simultaneously as discussed earlier using formulas (13) and (14). Values of t (time since last contact) and $\Omega_j(t)$ are retained for later use for a number of increments. The reason for retaining the time values is that the Runge-Kutta routine determines and adjusts its own step size, defying user control. Values of t, $P_{1j}(t)$, $P_{2j}(t)$, $P_{3j}(t)$, $P_{4j}(t)$, $P_{5j}(t)$, $Q_j(t)$, and $\Omega_j(t)$ are printed for a number of values of t.

The limits B_{ij} and L_j are calculated by formulas (27) and (28). The steady state contact type probabilities θ_j are calculated by formula (12). The B_{ij}, L_j, and θ_j are then printed out.

Values of the range distribution $G(R)$ are calculated from the tables of t, $\Omega_j(t)$ and printed for the values R = 5,10, 15,•••,200 nm., as discussed earlier using formula (29).

Peter S. Shoenfeld

Peter S. Shoenfeld

Reference [a]: L. D. Stone, Theory of Optimal Search, Academic Press, New York, 1975.

APPENDIX A

COMPUTER PROGRAM LISTING

```
      DIMENSION PRMT(5),DERP(5),P(5),AUX(8,5)
      DIMENSION PD(3,3),DPD(3,3),T(3,400),OMEGA(3,400)
      DIMENSION OMINF(3),THETA(3),B(3,3),N(3)
      REAL LAMBDA
      COMMON VS,ALPHA,R,LAMBDA,RHO,A01,A02,J,PD,DPD,T,OMEGA,
     2 OMINF,N,TLAST,PDLAST
      DATA PRMT/0.,400.,0.0625,0.0001,0./
      DATA NDIM/5/,TMAX/300./
C           READ PARAMETERS
5     TYPE 10
10    FORMAT(1X,'VS,RE,ALPHA,LAMBDA,RHO,AZERO1,AZERO3?'/)
      ACCEPT 20,VS,R,ALPHA,LAMBDA,RHO,A01,A02
20    FORMAT(7G)
      EXTERNAL FCT,OUTP
C           INTEGRATE O.D.E. SYSTEM FOR EACH CONTACT TYPE
C            BY RUNGE KUTTA METHOD USING SUBROUTINE RKGS
      PRMT(2)=TMAX
      DO 100 J=1,3
      PRMT(5)=0.
      DO 30 K=1,5
      P(K)=0.
30    DERP(K)=0.2
      IF(J.NE.2)P(4)=1.
      IF(J.EQ.2)P(5)=1.
      N(J)=0
      OMINF(J)=0.
      TLAST=0.
      PDLAST=0.
      TYPE 50,J
50    FORMAT(20X,'TYPE',I3,' CONTACT'//4X,'T', 5X,'P1', 4X,
     2 'P2', 4X,'P3', 4X,'P4', 4X,'P5', 4X,'PD',4X,'OMEGA'//)
      CALL RKGS(PRMT,P,DERP,NDIM,IHLF,FCT,OUTP,AUX)
100   TYPE 110,IHLF
110   FORMAT(1X,'IHLF=',I4//)
C           FIND LIMITS TO OBTAIN TRANSITION MATRIX B(I,J) AND
C           MEAN HOLDING TIMES OMINF(J)
      DO 200 J=1,3
      PDTOT=PD(J,1)+PD(J,2)+PD(J,3)
      DPDTOT=DPD(J,1)+DPD(J,2)+DPD(J,3)
      OMINF(J)=OMINF(J)+(1.-PDTOT)**2/DPDTOT
      DO 200 I=1,3
200   B(I,J)=PD(J,I)+DPD(J,I)*(1.-PDTOT)/DPDTOT
C           FIND EIGENVECTOR OF B(I,J) TO GET STEADY STATE PROBABILITIES
      D=(B(1,1)-1.)*(B(2,2)-1.)-B(1,2)*B(2,1)
      THETA(1)=(B(1,3)*(B(2,2)-1.)-B(2,3)*B(1,2))/D
      THETA(2)=(B(2,3)*(B(1,1)-1.)-B(1,3)*B(2,1))/D
      THETA(3)=-1.
      SUM=THETA(1)+THETA(2)+THETA(3)
      DO 250 J=1,3
250   THETA(J)=THETA(J)/SUM
```

```
C          PRINT THESE RESULTS
      TYPE 260,OMINF,THETA,((B(I,J),J=1,3),I=1,3)
260   FORMAT(//20X,'MEAN HOLDING TIMES'//
     2 3(3X,F13.5)//20X,'STEADY STATE PROBABILITIES'//3(3X,F13.5)//
     3 20X,'TRANSITION MATRIX'//3(3(3X,F13.5)/))
C          CALCULATE RANGE DISTRIBUTION AND PRINT
      TYPE 270
270   FORMAT(//1X,'RANGE',5X,'CUM.PROB.'//)
      DO 600 K=1,40
      RR=5.*K
      PROB=0.
      DO 500 J=1,3
      SUM=0.
      NN=0
      TOLD=0.
280   IF(NN.EQ.N(J))GO TO 400
      H1=T(J,NN+1)-TOLD
      IF(NN+2.LE.N(J))H2=T(J,NN+2)-T(J,NN+1)
      IF(H2.NE.H1.OR.NN+2.GT.N(J))GO TO 300
      SUM=SUM+H1*(ELT(NN,RR)+4.*ELT(NN+1,RR)+ELT(NN+2,RR))/3.
      TOLD=T(J,NN+2)
      NN=NN+2
      GO TO 280
300   SUM=SUM+H1*(ELT(NN,RR)+ELT(NN+1,RR))/2.
      TOLD=T(J,NN+1)
      NN=NN+1
      GO TO 280
400   PROB=PROB+THETA(J)*SUM
500   CONTINUE
      PROB=9.424778*RHO*RR*RR*PROB/(THETA(1)*OMINF(1)+
     2 THETA(2)*OMINF(2)+THETA(3)*OMINF(3))+
     3 1.-EXP(-9.424778*RR*RR/(RHO*TMAX))
      TYPE 510,RR,PROB
510   FORMAT(1X,F6.1,2X,F7.3)
600   CONTINUE
      GO TO 5
      END
      FUNCTION ELT(I,RR)
C        EVALUATES INTEGRAND FOR RANGE DISTRIBUTION CALCULATION
      DIMENSION PD(3,3),DPD(3,3),T(3,400),OMEGA(3,400),
     2 OMINF(3),THETA(3),B(3,3),N(3)
      REAL LAMBDA
      COMMON VS,ALPHA,R,LAMBDA,RHO,A01,A02,J,PD,DPD,T,OMEGA,
     2 OMINF,N,TLAST,PDLAST
      A0=A01
      IF(J.EQ.3)A0=A02
      IF(I.GT.0.AND.I.LE.N(J))GO TO 20
      ELT=0.
      GO TO 30
20    TT=T(J,I)
      ELT=EXP(-9.424778*RR*RR/(A0+RHO*TT))*OMEGA(J,I)/
     2 (A0+RHO*TT)**2
30    RETURN
      END
```

```
      SUBROUTINE FCT(TT,P,DERP)
C         EVALUATES DERIVATIVES FOR RUNGE-KUTTA ROUTINE
      DIMENSION P(5),DERP(5)
      DIMENSION PD(3,3),DPD(3,3),T(3,400),OMEGA(3,400),
     2 OMINF(3),THETA(3),B(3,3),N(3)
      REAL LAMBDA
      COMMON VS,ALPHA,R,LAMBDA,RHO,A01,A02,J,PD,DPD,T,OMEGA,
     2 OMINF,N,TLAST,PDLAST
      A(X)=A0+RHO*X
      BETA(X)=2.*R*VS/A(X)
      PP(X)=EXP(-3.141593*R*R/A(X))
      GAMMA(X)=(PP(X)/(1.-PP(X)))*(BETA(X)+3.141593*R*R*RHO/A(X)**2)
      A0=A01
      IF(J.EQ.3)A0=A02
      DERP(1)=ALPHA*P(4)
      DERP(2)=ALPHA*P(5)
      DERP(3)=LAMBDA*P(4)
      DERP(4)=-(ALPHA+LAMBDA+GAMMA(TT))*P(4)+BETA(TT)*P(5)
      DERP(5)=GAMMA(TT)*P(4)-(ALPHA+BETA(TT))*P(5)
      RETURN
      END
      SUBROUTINE OUTP(TT,P,DERP,IHLF,NDIM,PRMT)
C         OUTPUT ROUTINE FOR RUNGE-KUTTA INTEGRATION
      REAL LAMBDA
      DIMENSION P(5),DERP(5),PRMT(5)
      DIMENSION PD(3,3),DPD(3,3),T(3,400),OMEGA(3,400),
     2 OMINF(3),THETA(3),B(3,3),N(3)
      COMMON VS,ALPHA,R,LAMBDA,RHO,A01,A02,J,PD,DPD,T,OMEGA,
     2 OMINF,N,TLAST,PDLAST
      DO 10 I=1,3
      PD(J,I)=P(I)
10    DPD(J,I)=DERP(I)
      PDD=P(1)+P(2)+P(3)
      OMINF(J)=OMINF(J)+(TT-TLAST)*(1.-(PDD+PDLAST)/2.)
      TLAST=TT
      PDLAST=PDD
      IF(AMOD(TT,.25).GE.0.0001)GO TO 50
      IF(AMOD(TT,2.).GE.0.0001.AND.TT.GT.4.)GO TO 50
      IF(AMOD(TT,8.).GE.0.0001.AND.TT.GT.20.)GO TO 50
      IF(AMOD(TT,16.).GE.0.0001.AND.TT.GT.100.)GO TO 50
      TYPE 20,TT,P,PDD,OMINF(J)
20    FORMAT(1X,F7.2,6(1X,F5.3),1X,F6.2)
50    IF(AMOD(TT,.0625).GE.0.0001)GO TO 70
      IF(AMOD(TT,1.).GE.0.0001.AND.TT.GT.4.)GO TO 70
      N(J)=N(J)+1
      T(J,N(J))=TT
      OMEGA(J,N(J))=OMINF(J)
70    IF(PDD.GT.1.)PRMT(5)=1.
      RETURN
      END
*
```

```
C     ..................................................................
C
C        SUBROUTINE RKGS
C
C        PURPOSE
C           TO SOLVE A SYSTEM OF FIRST ORDER ORDINARY DIFFERENTIAL
C           EQUATIONS WITH GIVEN INITIAL VALUES.
C
C        USAGE
C           CALL RKGS (PRMT,Y,DERY,NDIM,IHLF,FCT,OUTP,AUX)
C           PARAMETERS FCT AND OUTP REQUIRE AN EXTERNAL STATEMENT.
C
C        DESCRIPTION OF PARAMETERS
C           PRMT   - AN INPUT AND OUTPUT VECTOR WITH DIMENSION GREATER
C                    OR EQUAL TO 5, WHICH SPECIFIES THE PARAMETERS OF
C                    THE INTERVAL AND OF ACCURACY AND WHICH SERVES FOR
C                    COMMUNICATION BETWEEN OUTPUT SUBROUTINE (FURNISHED
C                    BY THE USER) AND SUBROUTINE RKGS. EXCEPT PRMT(5)
C                    THE COMPONENTS ARE NOT DESTROYED BY SUBROUTINE
C                    RKGS AND THEY ARE
C           PRMT(1)- LOWER BOUND OF THE INTERVAL (INPUT),
C           PRMT(2)- UPPER BOUND OF THE INTERVAL (INPUT),
C           PRMT(3)- INITIAL INCREMENT OF THE INDEPENDENT VARIABLE
C                    (INPUT),
C           PRMT(4)- UPPER ERROR BOUND (INPUT). IF ABSOLUTE ERROR IS
C                    GREATER THAN PRMT(4), INCREMENT GETS HALVED.
C                    IF INCREMENT IS LESS THAN PRMT(3) AND ABSOLUTE
C                    ERROR LESS THAN PRMT(4)/50, INCREMENT GETS DOUBLED.
C                    THE USER MAY CHANGE PRMT(4) BY MEANS OF HIS
C                    OUTPUT SUBROUTINE.
C           PRMT(5)- NO INPUT PARAMETER. SUBROUTINE RKGS INITIALIZES
C                    PRMT(5)=0. IF THE USER WANTS TO TERMINATE
C                    SUBROUTINE RKGS AT ANY OUTPUT POINT, HE HAS TO
C                    CHANGE PRMT(5) TO NON-ZERO BY MEANS OF SUBROUTINE
C                    OUTP. FURTHER COMPONENTS OF VECTOR PRMT ARE
C                    FEASIBLE IF ITS DIMENSION IS DEFINED GREATER
C                    THAN 5. HOWEVER SUBROUTINE RKGS DOES NOT REQUIRE
C                    AND CHANGE THEM. NEVERTHELESS THEY MAY BE USEFUL
C                    FOR HANDING RESULT VALUES TO THE MAIN PROGRAM
C                    (CALLING RKGS) WHICH ARE OBTAINED BY SPECIAL
C                    MANIPULATIONS WITH OUTPUT DATA IN SUBROUTINE OUTP.
C           Y      - INPUT VECTOR OF INITIAL VALUES.  (DESTROYED)
C                    LATERON Y IS THE RESULTING VECTOR OF DEPENDENT
C                    VARIABLES COMPUTED AT INTERMEDIATE POINTS X.
C           DERY   - INPUT VECTOR OF ERROR WEIGHTS.  (DESTROYED)
C                    THE SUM OF ITS COMPONENTS MUST BE EQUAL TO 1.
C                    LATERON DERY IS THE VECTOR OF DERIVATIVES, WHICH
C                    BELONG TO FUNCTION VALUES Y AT A POINT X.
C           NDIM   - AN INPUT VALUE, WHICH SPECIFIES THE NUMBER OF
C                    EQUATIONS IN THE SYSTEM.
C           IHLF   - AN OUTPUT VALUE, WHICH SPECIFIES THE NUMBER OF
C                    BISECTIONS OF THE INITIAL INCREMENT. IF IHLF GETS
C                    GREATER THAN 10, SUBROUTINE RKGS RETURNS WITH
C                    ERROR MESSAGE IHLF=11 INTO MAIN PROGRAM. ERROR
C                    MESSAGE IHLF=12 OR IHLF=13 APPEARS IN CASE
C                    PRMT(3)=0 OR IN CASE SIGN(PRMT(3)).NE.SIGN(PRMT(2)-
C                    PRMT(1)) RESPECTIVELY.
```

```
C          FCT     - THE NAME OF AN EXTERNAL SUBROUTINE USED. THIS
C                    SUBROUTINE COMPUTES THE RIGHT HAND SIDES DERY OF
C                    THE SYSTEM TO GIVEN VALUES X AND Y. ITS PARAMETER
C                    LIST MUST BE X,Y,DERY. SUBROUTINE FCT SHOULD
C                    NOT DESTROY X AND Y.
C          OUTP    - THE NAME OF AN EXTERNAL OUTPUT SUBROUTINE USED.
C                    ITS PARAMETER LIST MUST BE X,Y,DERY,IHLF,NDIM,PRMT.
C                    NONE OF THESE PARAMETERS (EXCEPT, IF NECESSARY,
C                    PRMT(4),PRMT(5),...) SHOULD BE CHANGED BY
C                    SUBROUTINE OUTP. IF PRMT(5) IS CHANGED TO NON-ZERO,
C                    SUBROUTINE RKGS IS TERMINATED.
C          AUX     - AN AUXILIARY STORAGE ARRAY WITH 8 ROWS AND NDIM
C                    COLUMNS.
C
C       REMARKS
C          THE PROCEDURE TERMINATES AND RETURNS TO CALLING PROGRAM, IF
C          (1) MORE THAN 10 BISECTIONS OF THE INITIAL INCREMENT ARE
C              NECESSARY TO GET SATISFACTORY ACCURACY (ERROR MESSAGE
C              IHLF=11),
C          (2) INITIAL INCREMENT IS EQUAL TO 0 OR HAS WRONG SIGN
C              (ERROR MESSAGES IHLF=12 OR IHLF=13),
C          (3) THE WHOLE INTEGRATION INTERVAL IS WORKED THROUGH,
C          (4) SUBROUTINE OUTP HAS CHANGED PRMT(5) TO NON-ZERO.
C
C       SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED
C          THE EXTERNAL SUBROUTINES FCT(X,Y,DERY) AND
C          OUTP(X,Y,DERY,IHLF,NDIM,PRMT) MUST BE FURNISHED BY THE USER.
C
C       METHOD
C          EVALUATION IS DONE BY MEANS OF FOURTH ORDER RUNGE-KUTTA
C          FORMULAE IN THE MODIFICATION DUE TO GILL. ACCURACY IS
C          TESTED COMPARING THE RESULTS OF THE PROCEDURE WITH SINGLE
C          AND DOUBLE INCREMENT.
C          SUBROUTINE RKGS AUTOMATICALLY ADJUSTS THE INCREMENT DURING
C          THE WHOLE COMPUTATION BY HALVING OR DOUBLING. IF MORE THAN
C          10 BISECTIONS OF THE INCREMENT ARE NECESSARY TO GET
C          SATISFACTORY ACCURACY, THE SUBROUTINE RETURNS WITH
C          ERROR MESSAGE IHLF=11 INTO MAIN PROGRAM.
C          TO GET FULL FLEXIBILITY IN OUTPUT, AN OUTPUT SUBROUTINE
C          MUST BE FURNISHED BY THE USER.
C          FOR REFERENCE, SEE
C          RALSTON/WILF, MATHEMATICAL METHODS FOR DIGITAL COMPUTERS,
C          WILEY, NEW YORK/LONDON, 1960, PP.110-120.
C
C     ..................................................................
C
```

```
      SUBROUTINE RKGS(PRMT,Y,DERY,NDIM,IHLF,FCT,OUTP,AUX)
C
C
      DIMENSION Y(1),DERY(1),AUX(8,1),A(4),B(4),C(4),PRMT(1)
      DO 1 I=1,NDIM
    1 AUX(8,I)=.06666667*DERY(I)
      X=PRMT(1)
      XEND=PRMT(2)
      H=PRMT(3)
      PRMT(5)=0.
      CALL FCT(X,Y,DERY)
C
C     ERROR TEST
      IF(H*(XEND-X))38,37,2
C
C     PREPARATIONS FOR RUNGE-KUTTA METHOD
    2 A(1)=.5
      A(2)=.2928932
      A(3)=1.707107
      A(4)=.1666667
      B(1)=2.
      B(2)=1.
      B(3)=1.
      B(4)=2.
      C(1)=.5
      C(2)=.2928932
      C(3)=1.707107
      C(4)=.5
C
C     PREPARATIONS OF FIRST RUNGE-KUTTA STEP
      DO 3 I=1,NDIM
      AUX(1,I)=Y(I)
      AUX(2,I)=DERY(I)
      AUX(3,I)=0.
    3 AUX(6,I)=0.
      IREC=0
      H=H+H
      IHLF=-1
      ISTEP=0
      IEND=0
C
C
C     START OF A RUNGE-KUTTA STEP
    4 IF((X+H-XEND)*H)7,6,5
    5 H=XEND-X
    6 IEND=1
C
C     RECORDING OF INITIAL VALUES OF THIS STEP
    7 CALL OUTP(X,Y,DERY,IREC,NDIM,PRMT)
      IF(PRMT(5))40,8,40
    8 ITEST=0
    9 ISTEP=ISTEP+1
C
C
```

```
C     START OF INNERMOST RUNGE-KUTTA LOOP
      J=1
   10 AJ=A(J)
      BJ=B(J)
      CJ=C(J)
      DO 11 I=1,NDIM
      R1=H*DERY(I)
      R2=AJ*(R1-BJ*AUX(6,I))
      Y(I)=Y(I)+R2
      R2=R2+R2+R2
   11 AUX(6,I)=AUX(6,I)+R2-CJ*R1
      IF(J-4)12,15,15
   12 J=J+1
      IF(J-3)13,14,13
   13 X=X+.5*H
   14 CALL FCT(X,Y,DERY)
      GOTO 10
C     END OF INNERMOST RUNGE-KUTTA LOOP
C
C
C     TEST OF ACCURACY
   15 IF(ITEST)16,16,20
C
C     IN CASE ITEST=0 THERE IS NO POSSIBILITY FOR TESTING OF ACCURACY
   16 DO 17 I=1,NDIM
   17 AUX(4,I)=Y(I)
      ITEST=1
      ISTEP=ISTEP+ISTEP-2
   18 IHLF=IHLF+1
      X=X-H
      H=.5*H
      DO 19 I=1,NDIM
      Y(I)=AUX(1,I)
      DERY(I)=AUX(2,I)
   19 AUX(6,I)=AUX(3,I)
      GOTO 9
C
C     IN CASE ITEST=1 TESTING OF ACCURACY IS POSSIBLE
   20 IMOD=ISTEP/2
      IF(ISTEP-IMOD-IMOD)21,23,21
   21 CALL FCT(X,Y,DERY)
      DO 22 I=1,NDIM
      AUX(5,I)=Y(I)
   22 AUX(7,I)=DERY(I)
      GOTO 9
C
C     COMPUTATION OF TEST VALUE DELT
   23 DELT=0.
      DO 24 I=1,NDIM
   24 DELT=DELT+AUX(8,I)*ABS(AUX(4,I)-Y(I))
      IF(DELT-PRMT(4))28,28,25
C
```

```
C       ERROR IS TOO GREAT
   25 IF(IHLF-10)26,36,36
   26 DO 27 I=1,NDIM
   27 AUX(4,I)=AUX(5,I)
      ISTEP=ISTEP+ISTEP-4
      X=X-H
      IEND=0
      GOTO 18
C
C       RESULT VALUES ARE GOOD
   28 CALL FCT(X,Y,DERY)
      DO 29 I=1,NDIM
      AUX(1,I)=Y(I)
      AUX(2,I)=DERY(I)
      AUX(3,I)=AUX(6,I)
      Y(I)=AUX(5,I)
   29 DERY(I)=AUX(7,I)
      CALL OUTP(X-H,Y,DERY,IHLF,NDIM,PRMT)
      IF(PRMT(5))40,30,40
   30 DO 31 I=1,NDIM
      Y(I)=AUX(1,I)
   31 DERY(I)=AUX(2,I)
      IREC=IHLF
      IF(IEND)32,32,39
C
C       INCREMENT GETS DOUBLED
   32 IHLF=IHLF-1
      ISTEP=ISTEP/2
      H=H+H
      IF(IHLF)4,33,33
   33 IMOD=ISTEP/2
      IF(ISTEP-IMOD-IMOD)4,34,4
   34 IF(DELT-.02*PRMT(4))35,35,4
   35 IHLF=IHLF-1
      ISTEP=ISTEP/2
      H=H+H
      GOTO 4
C
C
C       RETURNS TO CALLING PROGRAM
   36 IHLF=11
      CALL FCT(X,Y,DERY)
      GOTO 39
   37 IHLF=12
      GOTO 39
   38 IHLF=13
   39 CALL OUTP(X,Y,DERY,IHLF,NDIM,PRMT)
   40 RETURN
      END
*
```